
The enlightment path to GADTs

I Evaluator with dynamic type checking.
I Well-typed expressions with tag-less evaluator.
I Sized lists and utility functions (head , tail , map, zipWith,

replicate, transpose).
I Simply typed lambda calculus
I If you are still alive, check out the work done by the

dependent types community and look for more examples.
The examples here are based on work by the Epigram and
Omega people.

An interpreter

I Write an interpreter for the following language:

data UExpr = LitInt Int
| LitBool Bool
| Inc UExpr
| IsZ UExpr
| If UExpr UExpr UExpr

I Values in the language:

data Val = VBool Bool
| VInt Int

An interpreter

I Interpreter:

eval :: UExpr → Val
eval (LitInt n) = VInt n
eval (Inc e)

= case eval e of
VInt n → VInt (n + 1)

→ error "inc applied to non-int"
eval (IsZ e)

= case eval e of
VInt n → VBool (n = = 0)

→ error "isZ applied to non-int"

An interpreter

I Interpreter:

eval (If c e1 e2)
= case eval c of

VBool b → if b then eval e1 else eval e2
→ error "if condition is non-bool"

I Evaluation of if isZ 1 then 1 else inc 2:

eval (If (IsZ (LitInt 1)) (LitInt 1) (Inc (LitInt 2)))
⇒ 3

I Evaluation of inc True:

eval (Inc (LitBool True))
⇒ error "inc applied to non-int"

Interpreter

I eval is a tag-ful interpreter: it does type checking at
runtime by checking the tags of values.

I If the expression is type-correct, the UExpr datatype cannot
express it.

I How can we encode well-typedness into the expression
datatype?

Well-typed expressions

I Expressions of different types can be encoded in different
datatypes:

data IExpr = ILit Int
| IInc IExpr
| IIf (IfExpr IExpr)

data BExpr = BLit Bool
| BIsZ IExpr
| BIf (IfExpr BExpr)

data IfExpr a = IfExpr BExpr a a

Well-typed expressions

I But now eval becomes several functions:

evalIExpr :: IExpr → Int
evalIExpr (ILit n) = n
evalIExpr (IInc e) = evalIExpr e + 1
evalIExpr (IIf e) = evalIExpr (evalIf e)

evalBExpr :: BExpr → Bool
evalBExpr (BLit b) = b
evalBExpr (BIsZ e) = evalIExpr e = = 0
evalBExpr (BIf e) = evalBExpr (evalIf e)

evalIf :: IfExpr a → a
evalIf (IfExpr c e1 e2) = if evalBExpr c then e1 else e2

Well-typed expressions

I Splitting datatypes splits the evaluator.
I Adding pairs or other datatypes becomes even more tricky

(more splitting).
I There are good reasons to keep expressions as one

datatype.
• we would like to have one evaluate function.
• it is difficult to write a function that parses expressions with

split types!
I However we learned two valuable lessons from enforcing

well-typedness in the datatype:
• we can omit dynamic type checking from the evaluator:

tag-less evaluation.
• the evaluator is more likely to be correct due to the more

precise types, i.e. the int expression evaluator cannot
return a bool!

I Now we’ll see how to encode well-typed expressions in
GADTs.

Generalised algebraic datatypes

I From Peyton-Jones, Vytiniotis and Weirich 2006:

data TExpr a where
TLitInt :: Int → TExpr Int
TIsZ :: TExpr Int → TExpr Bool
TIf :: TExpr Bool → TExpr a → TExpr a → TExpr a

I On the declaration syntax:
I Constructors are given type signatures, like functions.
I In the signature, the arguments are constructor fields.
I Constructors may restrict the type arguments in the return

type.

Generalised algebraic datatypes (GADTs)

I From Peyton-Jones, Vytiniotis and Weirich 2006:

data TExpr a where
TLitInt :: Int → TExpr Int
TIsZ :: TExpr Int → TExpr Bool
TIf :: TExpr Bool → TExpr a → TExpr a → TExpr a

I Here, the argument to TExpr is called a type index.
I The type index tells something about the constructors that

conform the value.
I At the same time the constructors use indices to restrict

the values that can be plugged in as fields.
I In this case TExpr represents only well-typed expressions

(modulo ⊥ values).

Interpreter for well-typed expressions

I A tag-less interpreter from Peyton-Jones, Vytiniotis and
Weirich 2006:

evalT :: TExpr a → a
evalT (TLitInt i) = i
evalT (TIsZ e) = evalT e = = 0
evalT (TIf c e1 e2) = if evalT c then evalT e1 else evalT e2

I Here’s how typing works: the TExpr a in the signature is
refined to the constructor return type in every arm.

I So, for TIsZ the right hand side must return TExpr Bool
rather than TExpr a.

I The type checking algorithm requires a type signature for
the function.

More GADT examples

I In this lecture we will see GADT examples that:
• don’t need function cases that are ruled out by datatype

properties.
• are more reliable in the sense that functions cannot break

datatype properties.
I The examples that we will see are:

• Sized lists.
• Simply typed lambda calculus.

Sized lists

I Sized lists have a type index that gives the number of
elements.

I We first define type-level naturals to count elements:

data Zero
data Suc a

I Sized lists:

data List a sz where
Nil :: List a Zero
Cons :: a → List a sz → List a (Suc sz)

Sized lists examples

I Examples:

ex1 = Nil :: List a Zero
ex2 = Cons 1 Nil :: List Int (Suc Zero)
ex3 = Cons 1 (Cons 2 Nil) :: List Int (Suc (Suc Zero))

head and tail

I We can make head and tail total because empty lists are
not part of their domain:

head :: List a (Suc sz) → a
head (Cons x xs) = x
tail :: List a (Suc sz) → List a sz
tail (Cons x xs) = xs

I If you go ahead and give a case for Nil :

head Nil = error "impossible"

the compiler will complain:
Inaccessible case alternative: Can’t match types ‘Zero’ and ‘Suc n’
In the pattern: Nil
In the definition of ‘head’: head Nil = error "impossible"

map and zipWith

I The type guarantees that the result list has the same
length as the initial one.

map :: (a → b) → List a sz → List b sz
map f Nil = Nil
map f (Cons x xs) = Cons (f x) (map f xs)

I For zipWith the type makes the case for different lengths
unnecessary.

zipWith :: (a → b → c) → List a sz → List b sz → List c sz
zipWith f Nil Nil

= Nil
zipWith f (Cons x xs) (Cons y ys)

= Cons (f x y) (zipWith f xs ys)



17

map and zipWith

I The type guarantees that the result list has the same
length as the initial one.

map :: (a → b) → List a sz → List b sz
map f Nil = Nil
map f (Cons x xs) = Cons (f x) (map f xs)

I For zipWith the type makes the case for different lengths
unnecessary.

zipWith :: (a → b → c) → List a sz → List b sz → List c sz
zipWith f Nil Nil

= Nil
zipWith f (Cons x xs) (Cons y ys)

= Cons (f x y) (zipWith f xs ys)



18

replicate

I For replicate we will need value level naturals

replicate :: RNat sz → a → List a sz

I so we define:

data RNat n where
RZero :: RNat Zero
RSuc :: RNat sz → RNat (Suc sz)

I And now we are able to write replicate down:

replicate RZero x = Nil
replicate (RSuc n) x = Cons x (replicate n x)

transpose

I The type of transpose (drawing):

transpose :: List (List a cols) rows → List (List a rows) cols

I The base case is tricky, transposing 0× 3 gives 3× 0.

transpose Nil = replicate ___ Nil

I So we need to pass the number of columns as an
argument:

transpose :: RNat cols
→ List (List a cols) rows → List (List a rows) cols

transpose ncols Nil = replicate ncols Nil
transpose ncols (Cons xs xss)

= zipWith Cons xs (transpose ncols xss)

transpose

I We can avoid passing the number of columns explictly in
two ways:

• write a transpose that requires non-zero dimensions.
• use a type-class Nat that reifies type naturals to the value

level, use a method reifyNat :: Nat sz ⇒ RNat sz.
I These are exercises for you. Since you have to code this,

you might as well continue with matrix multiplication.

Preliminary: existential types

I Suppose we want to obtain sized lists from normal lists, as
follows:

toSized :: [a] → List a sz

What is sz supposed to be?!?!
I A non-solution is to assume that sz is universally

quantified. But this is wrong! Because

toSized [ ] :: List Int (Suc Zero)

is now well typed!
I The solution is to use an existential quantified variable to

say that we just don’t know what sz is at compile time:

toSized :: [a] → (∃sz . List a sz)

Preliminary: existential types

I With GHC, we can only specify existential variables in data
types:

data ToSizedRes a where
ToSizedRes :: List a sz → ToSizedRes a

ToSizedRes wraps a sized list and "forgets" the size of it.
Now we can define toSized .

toSized :: [a] → ToSizedRes a
toSized [ ] = ToSizedRes Nil
toSized (x : xs)

= case toSized xs of
ToSizedRes ls → ToSizedRes (Cons x ls)

Preliminary: existential types

I But now, what is the point of having a sized list and not
knowing the size?

funnyHead :: [a] → a
funnyHead ls

= case toSized ls of
ToSizedRes sls → head sls -- type error!!!

I This problem with existentials is solved in two ways:
• wrapping something that knows what to do with the

existential, or
• wrapping something that tells us what the existential is, i.e.

a witness of the existential.
I We follow the second solution.

Preliminary: existential types

I ToSizedRes packages a witness of the list size.

data ToSizedRes a where
ToSizedRes :: RNat sz → List a sz → ToSizedRes a

toSized :: [a] → ToSizedRes a
toSized [ ] = ToSizedRes RZero Nil
toSized (x : xs)

= case toSized xs of
ToSizedRes wsz ls

→ ToSizedRes (RSuc wsz) (Cons x ls)

Preliminary: existential types

I Now we use the witness:

funnyHead :: [a] → a
funnyHead ls

= case toSized ls of
ToSizedRes (RSuc ) sls → head sls
→ error "funnyHead []"

I The witness seems redudant in this case, but if you have
something more complex than a list, it starts to pay off.

Preliminary: existential types

I Challenge: Implement a UExpr to TTerm converter
(untyped to well-typed expressions).

I Here a witness for the existential will be even more useful.

append

I append is a challenge because it would seem that its
signature requires evaluation at the type level.

append :: List a sz1 → List a sz2 → List a (plus sz1 sz2)

But plus is not definable in Haskell (unless you use some
serious type class hackery).

I However, we can encode this using existential types and a
witness that acts as evidence (proof) that sz3 = sz1 + sz2:

data Plus sz1 sz2 sz3 where
PlusBase :: Plus Zero sz2 sz2
PlusStep :: Plus sz1 sz2 sz3

→ Plus (Suc sz1) sz2 (Suc sz3)

append

I append returns the concatenated list and the evidence that
it has the size of the two input lists combined.

data AppendRes a sz1 sz2 where
AppendRes :: Plus sz1 sz2 sz3 → List a sz3
→ AppendRes a sz1 sz2

I The definition of append:

append :: List a sz1 → List a sz2 → AppendRes a sz1 sz2
append Nil ys = AppendRes PlusBase ys
append (Cons x xs) ys

= case append xs ys of
AppendRes proof res
→ AppendRes (PlusStep proof ) (Cons x res)

Simply typed lambda calculus

I Write an interpreter for the following language:

data Expr = Var String
| App Expr Expr
| Abs String Expr

and many constants and functions.
I The same problem as before, the evaluator must

distinguish between functions from non-functions and from
constants of different types.

I As before, we eliminate runtime type checking encoding
only well-typed terms.

Simply typed lambda calculus

I We say that the expression has type t under an
environment env

data Expr env t where

The evaluation will require an environment of an
appropriate type:

evaluate :: Env env → Expr env t → t

Simply typed lambda calculus

I Let us see pseudo-types for a few example expressions:

x :: Expr ((x :: a) : env) a
x y :: Expr ((x :: a → b) : (y :: b) : env) b
(λx . x) :: Expr env (a → a)

I Replacing names by (de Bruijn) indices we get:

0 :: Expr (a, env) a
0 1 :: Expr (a → b, (b, env)) b
(λ . 0) :: Expr env (a → a)

Simply typed lambda calculus

I Variables with type t under an environment env

data Var env t where
VFirst :: Var (a, env) a
VNext :: Var env a → Var (b, env) a

I Environments (heterogeneous list):

data Env env where
EEmpty :: Env ()
EExt :: a → Env env → Env (a, env)

I And variable lookup:

lookup :: Var env a → Env env → a
lookup VFirst (EExt x es) = x
lookup (VNext v) (EExt x es) = lookup v es

Simply typed lambda calculus

I Well-typed expressions

data Expr env t where
EVar :: Var env t → Expr env t
EApp :: Expr env (t1 → t2) → Expr env t1 → Expr env t2
EAbs :: Expr (a, env) b → Expr env (a → b)
ECon :: a → Expr env a

I Evaluator:

evaluate :: Env env → Expr env t → t
evaluate env (EVar v) = lookup v env
evaluate env (EApp e1 e2) = (evaluate env e1)

(evaluate env e2)
evaluate env (EAbs e) = λx → evaluate (EExt x env) e
evaluate env (ECon c) = c

Conclusions

I Increased reliability: functions cannot break
GADT-encoded properties.

I Properties not totally guaranteed: we can fake properties
with ⊥, and patterns are not properly checked for
exhaustiveness.

I Efficiency problems: Sometimes proof objects have to be
inspected at runtime.

I Existential syntax: GADT programs need a lot of them, it
might be time to revisit the "boxed existentials" design
choice.

I When is it useful for real programs? When you want to be
(almost) sure that a core part of your system is correct. I
have used them in such a way.


